
The augmented link also affects other links in the network and the activities of
generators at other nodes in the network. For example, as in Scenario 1, at peak
load all plants operate and node N2 supplies electricity into the network.
However, the amount that node N2 supplies at peak load is less than the amount
it supplies in Scenario 1, since more can be supplied by the lower cost plants at
nodes N1 and N4. Similarly, the amount that node N2 supplies at intermediate
load is smaller than the amount it supplies in the base case. Although node N2
still supplies electricity to node N3, this electricity is not all produced at N2.
Rather, node N2 produces some electricity but also receives electricity from
node N1 to be supplied indirectly to node N3. Finally, at base load, node N4
can now supply more electricity directly to node N3 than it could in the base
case, which lessens the amount that is supplied to node N3 from node N1.

Merit order for plant dispatch

The modelling also provides the merit order for dispatch of generators at the
generating nodes under the different scenarios (see figure 5.6).

In both scenarios, power stations are dispatched in order of marginal cost at the
node — that is from least cost to high cost. For example, at node N1, the
base/intermediate plant node, supplies electricity over all loads. Here the low
cost power stations, P2coal1 and P3coal1, are dispatched first then P1coal1 is
dispatched and finally the higher cost power station, P4gasol, is dispatched.
Node N4, another base/intermediate plant node, follows a similar pattern.
Finally, at node N2, peaking plant is not dispatched at base load, but is
dispatched in peak periods.

The augmentation of the transmission system affects both the timing of dispatch
of various plants and the amount of power produced by plants at all nodes in the
network. At node N4, the total supply increases in all periods and the mix in
supply by individual generators changes. For example, after augmentation,
generator P10coal1 supplies electricity in all periods (as it did in Scenario 1),
but the amount it supplies increases and P12gasol only supplies after
augmentation. The effect on generation at nodes directly effected by the link is
also seen at node N1. Here the node becomes a net importer in all time periods
— while in Scenario 1 it was a net importer at peak loads and a net exporter at
base loads — drawing from the increased output of node N4. As discussed
earlier, augmentation also affects nodes that are not directly tied to the link. For
example after augmentation, node N2 still acts as a peaking plant node, but it is
dispatched later, and supplies less electricity than it did in Scenario 1.

Figure 5.6: Merit order for plant dispatch, various scenarios

The income earned by the transmission system is given by the merchandising
surplus — this is the difference between the value of sales of electricity sold by
the network (exports) and the value of sales purchased the network (imports).
The source of this income is the shadow cost or imputed cost congestion from
transmission lines that are constraining the network and the rent earned on
transmission because marginal losses are higher than average losses.

Chao and Peck (1996) propose a system to allocate these rents to each
transmission line, taking into account the network externality caused by
Kirchoff’s laws. This method takes into account that injecting power or
withdrawing power at any node make a contribution to the whole systems
marginal losses and any transmission constraints on links.

Although we have been able to replicate the method published by Chao and
Peck (1996) using our modelling framework, we have not been able to replicate
it for the example presented here. We suspect that there may be an
intertemporal dimension introduced here because our example has 34 load
periods which are interdependent via generation. In the Chao and Peck (1996)
approach, the difference between nodal prices is decomposed into three
components, representing contributions to system congestion costs, marginal
transmission losses and the increase in energy purchased because energy losses.

5.10 Conclusion and further research

The model presented here provides insights into economic issues arising in
electricity markets. It has integrated demand, transmission and generation into a
single model to simulate the economically efficient operation of an electricity
market.

The framework could be developed to further explore pricing issues in
electricity markets. For example, introducing transmission capacity along each
node as a variable, including its annualised fixed cost in the objective function.
This would make transmission capacity along each node endogenous in the
model, thereby allowing for simultaneous optimisation of demand, transmission
capacity and generation.

Other examples are to extend to a multi-year model to evaluate long term
dynamic effects on the market, such as the timing of transmission and
generation augmentation, or investigate pricing rules to determine transmission
fees along each link that are consistent with an economically efficient market,
taking into account the network externality. This is an extension of the research
by Chao and Peck (1996).

Table 5.3: Income earned by the whole transmission system in each
load block ($m)

Scenario 1 Scenario 2
Load
block

Total Congestion Losses Total Congestion Losses

B1 0.74 0.11 0.63 1.06 0 1.06
B2 1.51 1.32 0.19 0.33 0 0.33
B3 1.98 1.65 0.33 0.57 0 0.57
B4 0.43 0 0.43 0.73 0 0.73
B5 0.74 0 0.74 1.24 0 1.24
B6 0.74 0.23 0.51 0.86 0 0.86
B7 0.76 0 0.76 1.28 0 1.28
B8 8.76 7.83 0.93 1.63 0 1.63
B9 0.94 0 0.94 1.57 0 1.57
B10 1.24 0 1.24 2.07 0 2.07
B11 1.52 0 1.52 2.52 0 2.52
B12 2.34 0 2.34 3.89 0 3.89
B13 1.97 0 1.97 3.27 0 3.27
B14 11.6 9.59 2 3.49 0 3.49
B15 2.58 0 2.58 4.46 0 4.46
B16 2.42 0 2.42 4.29 0 4.29
B17 13.65 11.94 1.71 3.24 0 3.24
B18 7.44 6.11 1.33 2.51 0 2.51
B19 4.53 3.4 1.13 2.13 0 2.13
B20 3.73 2.81 0.92 1.58 0 1.58
B21 3.37 2.55 0.82 1.26 0 1.26
B22 3.31 2.51 0.8 1.12 0 1.12
B23 3.2 2.43 0.77 0.98 0 0.98
B24 3.19 2.41 0.77 0.81 0 0.81
B25 0.64 0 0.64 0.83 0 0.83
B26 0.31 0 0.31 0.48 0 0.48
B27 0.35 0 0.35 0.63 0 0.63
B28 0.26 0 0.26 0.47 0 0.47
B29 0.2 0 0.2 0.45 0 0.45
B30 0.26 0 0.26 0.58 0 0.58
B31 0.2 0 0.2 0.45 0 0.45
B32 0.14 0 0.14 0.32 0 0.32
B33 0.1 0 0.1 0.24 0 0.24
B34 0.07 0 0.07 0.18 0 0.18
Total 85.22 54.89 30.31 51.52 0 51.52

Appendix A: GAMS code for the model
$offsymlist offsymxref
OPTIONS DECIMALS = 5;
OPTIONS LIMCOL =0;
OPTIONS LIMROW =0;
*OPTIONS SOLPRINT = OFF;
OPTIONS NLP = MINOS5;
OPTIONS SYSOUT = OFF;
OPTIONS ITERLIM = 20000;
OPTIONS RESLIM = 8000;

SETS
B LOAD BLOCKS
  / B1 * B34 /
N NODES
 / N1 * N4 /;
ALIAS (B,BP), (N,NP);

SETS
GR(N) NODES WHERE THERE IS GENERATION
  / N1, N2, N4  /

P GENERATING PLANTS
  / P1COAL1, P2COAL1, P3COAL1, P4GASOL, P5GASOC, P6GASCC,
    P7DIST, P8COAL1, P9COAL1, P10COAL1, P11COAL1, P12GASOL/
NG(N,P) THE PLANTS AT EACH NODE
  /N1.(P1COAL1, P2COAL1, P3COAL1, P4GASOL)
   N2.(P5GASOC, P6GASCC)
   N3.(P7DIST)
   N4.(P8COAL1, P9COAL1, P10COAL1, P11COAL1, P12GASOL)/
D(N) NODES WHERE THERE IS DEMAND
  / N1, N3 /
LK(N,NP) LINK BETWEEN EACH NODE IN THE NETWORK
  /N1.(N2, N3, N4)
   N2.(N1, N3)
   N3.(N1, N2, N4)
   N4.(N1, N3)/
NT(N) NODES THAT ARE NOT NODE N4
 /N1, N2, N3/

LABELS NAMES TO IDENTIFY PLANT DATA
  / UNITS    NO OF UNITS AT EACH PLANT
    AVAIL    AVAILABILITY OF EACH PLANT
    GENMIN   MINIMUM CAPACITY OF GENERATORS
    GENMAX   MAXIMUM CAPACITY OF GENERATORS
    FUELCOST FUELCOSTS ($ per MWh)
    CAPCOST  CAPITAL COST ($mill PER MW)
    LIFE     LIFE OF UNITS (years)/;

* ELEMENTS UNDERLYING DEMAND AT EACH NODE
TABLE DMD(B,N) DATA FOR LOAD AT EACH NODE IN EACH BLOCK (MW)
       N1       N3
B1     6500     2909.29
B2     6400     2814.83
B3     6300     2795.82
B4     6200     2785.80
B5     6100     2761.74
B6     6000     2724.57
B7     5900     2696.79

B8     5800     2644.10
B9     5700     2646.52
B10    5600     2617.59
B11    5500     2609.79
B12    5400     2600.58
B13    5300     2541.84
B14    5200     2472.71
B15    5100     2485.65
B16    5000     2436.29
B17    4900     2370.75
B18    4800     2321.46
B19    4700     2303.01
B20    4600     2247.04
B21    4500     2188.80
B22    4400     2149.72
B23    4300     2074.14
B24    4200     2029.16
B25    4100     1979.27
B26    4000     1912.78
B27    3900     1858.47
B28    3800     1794.04
B29    3700     1742.41
B30    3600     1676.24
B31    3500     1633.05
B32    3400     1583.99
B33    3300     1538.78
B34    3200     1463.29   ;

TABLE ENERGY(B,N) DATA FOR ENERGY AT EACH NODE IN EACH BLOCK (GWh)
       N1         N3
B1     246.42     107.45
B2     84.02      36.69
B3     151.70     66.82
B4     203.48     90.79
B5     360.99     162.01
B6     275.79     124.31
B7     438.85     199.21
B8     609.98     275.74
B9     624.37     287.49
B10    920.31     426.53
B11    1241.37    584.02
B12    2107.88    1005.71
B13    2195.53    1043.74
B14    2702.30    1273.22
B15    3400.93    1641.72
B16    3249.15    1566.77
B17    2432.17    1164.07
B18    2254.98    1079.34
B19    2094.40    1015.72
B20    1716.95    829.93
B21    1551.76    746.60
B22    1509.13    728.60
B23    1454.50    693.97
B24    1458.99    696.56
B25    1497.75    713.83
B26    854.03     403.11
B27    1114.64    524.91
B28    820.00     381.98
B29    798.85     370.99

B30    1007.68    462.51
B31    772.28     354.80
B32    532.17     244.34
B33    393.11     180.53
B34    294.14     133.52     ;

PARAMETERS
HOURS(B) NUMBER OF HOURS IN EACH LOAD BLOCK
/ B1      36.935
  B2      13.036
  B3      23.899
  B4      32.589
  B5      58.661
  B6      45.625
  B7      73.869
  B8     104.286
  B9     108.631
  B10    162.946
  B11    223.780
  B12    386.726
  B13    410.625
  B14    514.911
  B15    660.476
  B16    643.095
  B17    491.012
  B18    464.940
  B19    441.042
  B20    369.345
  B21    341.101
  B22    338.929
  B23    334.583
  B24    343.274
  B25    360.655
  B26    210.744
  B27    282.440
  B28    212.917
  B29    212.917
  B30    275.923
  B31    217.262
  B32    154.256
  B33    117.321
  B34     91.250 /;

PARAMETERS
PRI(B) BASE PRICE (cent per kWh)
/ B1     6.55
  B2     6.46
  B3     6.42
  B4     6.28
  B5     6.23
  B6     6.00
  B7     5.92
  B8     5.84
  B9     5.76
  B10    5.58
  B11    5.43
  B12    5.35
  B13    5.15
  B14    4.85
  B15    4.43

  B16    4.23
  B17    4.15
  B18    4.07
  B19    4.06
  B20    3.97
  B21    3.66
  B22    3.50
  B23    3.28
  B24    2.85
  B25    2.65
  B26    2.45
  B27    1.87
  B28    1.79
  B29    1.62
  B30    1.54
  B31    1.46
  B32    1.38
  B33    1.21
  B34    0.98 /;

PARAMETERS
 PRICE(N,B) PRICE AT EACH NODE IN EACH LOADBLOCK (cent per kwh);
 PRICE('N1',B) = 1.75*PRI(B);
 PRICE('N3',B) = 2.5*PRI(B);

PARAMETERS
  BETA(N)  PRICE ELASTICITIES OF DEMANDS IN EACH BLOCK AT EACH NODE
  IBETA(N,B) SLOPE IN INVERSE DEMAND FUNCTION IN EACH BLOCK AT EACH NODE
  ALPHA(N,B) CONSTANTS FOR INVERSE DEMAND FUNCTION IN EACH BLOCK AT EACH
NODE ($M PER PWH);
    BETA(N) $d(n)  = -0.3;
    IBETA(N,B) $d(n) = 1/BETA(N)*PRICE(N,B)*10/(ENERGY(B,N)/1000);
    ALPHA(N,B) $d(n) = PRICE(N,B)*10-IBETA(N,B)*ENERGY(B,N)/1000;

* ELEMENTS UNDERLYING SUPPLY AT EACH NODE
TABLE PDATA(P, LABELS) DATA FOR EACH OF THE GENERATING PLANTS
        UNITS AVAIL  GENMIN  GENMAX  FUELCOST  CAPCOST  LIFE
*                    MW      MW      $-MWh    $M-MW      years
P1COAL1     2    1   286     635     14.8     1.2        30
P2COAL1     3    1   110     287     14.0     1.4        30
P3COAL1     4    1   176     635     14.3     1.3        30
P4GASOL     2    1   200     250     18.5     1.2        30
P5GASOC     6    1   0       999     32       0.5        30
P6GASCC     6    1   0       999     23.5     0.85       25
P7DIST      4    1   0       999     150      0.3        30
P8COAL1     2    1   275     634     12.1     1.45       30
P9COAL1     2    1   227     480     13.0     1.3        30
P10COAL1    2    1   250     634     12.2     1.4        30
P11COAL1    2    1   205     445     14.8     1.25       30
P12GASOL    2    1   200     250     18.5     0.8        30;

SCALARS
  RHO INTEREST RATE / 0.08 /
  PRR PEAK RESERVE REQUIREMENT / 0.14 /

PARAMETERS
  PLANTCOST(P) FIXED ($M per GW) COSTS
  OPCOST(P,B) VARIABLE ($M per GW) COSTS;
  PLANTCOST(P) = (RHO/(1-(1+RHO)**(-PDATA(P,'LIFE'))))
                      *PDATA(P,'CAPCOST')*1000;
  OPCOST(P,B) = (PDATA(P,'FUELCOST'))/1000*HOURS(B);

* DISPLAY OPCOST, PLANTCOST;

* ELEMENTS UNDERLYING THE NETWORK
PARAMETERS
  R(N,NP) RESISTANCE ALONG LINE FROM NODE N TO NODE NP (OHMS)
  X(N,NP) INDUCTANCE ALONG LINE FROM NODE N TO NODE NP (OHMS)
  Y(N,NP) PARAMENTER IN KIRCHOFF EQUATON
  G(N,NP) PARAMETER IN KIRCHOFF EQUATION;

TABLE V(N,NP) VOLTAGE AMPLITUDE NODE N NODE NP (KV)
         N1   N2   N3   N4
N1       0    330  500  330
N2       330  0    330  0
N3       500  330  0    330
N4       330  0    330  0;

TABLE NUMBER(N,NP) NUMBER OF POWERLINES BETWEEN NODES IN THE NETWORK
         N1  N2   N3   N4
N1       0   6    4    2
N2       6   0    5    0
N3       4   5    0    2
N4       2   0    2    0;

TABLE DIST(N,NP) DISTANCE BETWEEN NODES IN THE NETWORK (KM)
         N1  N2   N3   N4
N1       0   150  375  150
N2       150  0   450  0
N3       375  450 0    400
N4       150  0   400  0;

TABLE RES(N,NP) RESISTANCE PER KM ALONG LINE FROM NODE N TO NODE NP (OHMS
per KM)
        N1    N2   N3    N4
N1      0     0.03 0.025 0.03
N2      0.03  0    0.03  0
N3      0.025 0.03 0     0.03
N4      0.03  0    0.03  0;

TABLE IMPED(N,NP) IMPEDANCE PER KM OF ALONG LINE FROM NODE N TO NODE NP
(OHMS per KM)
         N1   N2   N3   N4
N1       0    0.3  0.25 0.3
N2       0.3  0    0.3  0
N3       0.25 0.3  0    0.3
N4       0.3  0    0.3  0;

TABLE GRIDCAP(N,NP) ESTIMATED GRID CAPACITY IN THE NETWORK (MW)
         N1   N2   N3   N4
N1       0    3000 1500 1500
N2       3000 0    3000 0
N3       1500 3000 0    2000
N4       1500 0    2000 0 ;

X(N,NP)$lk(n,np) = IMPED(N,NP)*DIST(N,NP);
R(N,NP)$lk(n,np) = RES(N,NP)*DIST(N,NP);
Y(N,NP)$lk(n,np) = X(N,NP)/(SQR(R(N,NP))+SQR(X(N,NP)));
G(N,NP)$lk(n,np) = R(N,NP)/(SQR(R(N,NP))+SQR(X(N,NP)));

* THE MODEL

POSITIVE VARIABLES
  QS(N,B) ELECTRICTY SUPPLIED AT NODE N IN LOAD BLOCK B   (GW)
  QD(N,B) ELECTRICITY DEMANDED AT NODE N IN LOAD BLOCK B  (PWH)
  QGO(P,B) OUTPUT OF PLANT P IN LOAD BLOCK B (GW)
  QGC(P) OPERATING CAPACITY OF EACH PLANT P (GW)

FREE VARIABLES
  NSR NET SOCIAL WELFARE ($M)
  QP(N,NP,B) QUANTITY OF POWER FLOW AT EACH NODE IN EACH TIME BLOCK (GW)
  THETA(N,B) VOLTAGE phase ANGLES (RADIANS);

EQUATIONS
  OBJ NET SOCIAL WELFARE ($M)
  GENBAL(P,B) CAPACITY-OUTPUT BALANCE AT EACH GENERATOR IN EACH LOAD
BLOCK(GW)
  GMAXCAP(P) MAXIMUM CAPACITY CONSTRAINT FOR EACH PLANT(GW)
  NODEBAL(N,B) SUPPLY BALANCE AT EACH GENERATING NODE(GW)
  POWER(N,NP,B) POWER FLOW AT EACH NODE (GW)
  FLOWBAL(N,B) POWER FLOW BALANCE(GW)
  FLOWMAX(N,NP,B) MAXIMUM FLOW BETWEEN TWO NODES FOR EACH LOAD BLOCK(GW);

*objective function
OBJ..
  NSR =E= SUM((N,B)$d(n),(ALPHA(N,B)*QD(N,B)+0.5*IBETA(N,B)*(QD(N,B)**2)))
        - SUM((P,B),OPCOST(P,B)*QGO(P,B))
        - SUM((P),PLANTCOST(P)*QGC(P));

* IF THE PLANT IS SWITCHED ON, THE AMOUNT SUPPLIED IN EACH LOAD BLOCK
EQUALS THE AMOUNT SWITCHED ON
GENBAL(P,B)..
  QGO(P,B) =L= QGC(P)*PDATA(P,'AVAIL');

* THE OPERTING CAPACITY IS LESS THE THE UPPER BOUND ON GENERATION
GMAXCAP(P)..

  QGC(P) =L= PDATA(P,'UNITS')*PDATA(P,'GENMAX')/1000;

* THE NODE SUPPLY = THE TOTAL OUTPUT OF ALL GENERATORS AT THE NODE
NODEBAL(N,B)$GR(N)..
  QS(N,B) =L= SUM(P$(NG(N,P)), QGO(P,B));

* GETTING AN EQUATION FOR POWER
POWER(N,NP,B)$LK(N,NP)..
  QP(N,NP,B)*1000
=E= G(N,NP)*(V(N,NP)**2-V(N,NP)*V(NP,N))
+Y(N,NP)*V(N,NP)*V(NP,N)*THETA(N,B)$NT(N)
-Y(N,NP)*V(N,NP)*V(NP,N)*THETA(NP,B)$NT(NP)
+0.5*G(N,NP)*V(N,NP)*V(NP,N)*(THETA(N,B)$NT(N))**2
-0.5*G(N,NP)*V(N,NP)*V(NP,N)*2*THETA(N,B)$NT(N)*THETA(NP,B)$NT(NP)
+0.5*G(N,NP)*V(N,NP)*V(NP,N)*(THETA(NP,B)$NT(NP))**2;

* NODE DEMAND = OWN NODE SUPPLY - POWER FLOW FROM AT NODE FROM OTHER NODES
* A POSITIVE POWER FLOW AT THE NODE IS A FLOW AWAY FROM THE NODE
* A NEGATIVE POWER FLOW AT THE NODE IS A FLOW INTO THE NODE
FLOWBAL(N,B)..
  QD(N,B)$d(n)*(1000/HOURS(B))
         =E=  QS(N,B)$GR(N)
                   - SUM(NP$lk(N,NP), NUMBER(N,NP)*QP(N,NP,B));

* NET FLOW ALONG LINK < MAXIMUM FLOW ALONG LINK
FLOWMAX(N,NP,B)$LK(N,NP)..
  NUMBER(N,NP)*QP(N,NP,B)  =L= GRIDCAP(N,NP)/1000;

MODEL NETWORK /ALL/ ;

NETWORK.OPTFILE = 1 ;
* ABORT $(ALPHA("N1","B1") GT 0) "END";
SOLVE NETWORK MAXIMISING NSR USING NLP;

DISPLAY QD.L, QS.L, QGO.L, QGC.L, QP.L;
DISPLAY QD.M, QS.M, QGO.M, QGC.M, QP.M;
DISPLAY THETA.L;

